

## 第2講 プログラミングでLEDを制御する

Processing と Arduino を接続し LED を制御する。スイッチの ON/OFF を取得する。

### 1 Arduino の準備

本実習ではスイッチやセンサ、LEDなどをPCから制御するために Arduino<sup>1</sup> というマイコンを用います。

本来、Arduino と Processing を連携させるためには、シリアル通信を用いますが、そのためのプログラムを書くのは少し面倒です。実習では楽をするために、Firmata<sup>2</sup> を用います。

#### Todo

- Arduino IDE のインストール
- Arduino ドライバのインストール
- Arduino に Firmata を書き込む
- Processing に Arduino ライブラリをインストール
- 動作確認

### 2 Processing から Arduino を制御する

今回は Arduino の Digital Output を用いて LED の制御したり、Digital Input を用いてスイッチの ON/OFF を取得したりします。

まず、Processing から Arduino を用いるための準備をしましょう。

```
import processing.serial.*;
import cc.arduino.*;

Arduino arduino;

void setup() {
  // Arduino の初期化
  // シリアルポートの指定など
  // Arduino.list()[0] は環境によって変える
  arduino = new Arduino(this, Arduino.list()[0], 57600);
}
```

Arduino を使うためにはシリアルポートの指定をしなければなりません。Windows ならデバイスマネージャから確認することができます。Arduino.list()[0] のところを各自変更してください。

これらの命令は今後も Arduino を用いる際に必ず使うので忘れないように。

---

<sup>1</sup><http://arduino.cc>

<sup>2</sup><http://firmata.org>

## LED を点滅させる

Digital Output を使って LED を点滅させてみましょう。

### 回路を組む

LED を接続したピンに電圧をかけると LED が点灯する回路を作りましょう。

#### ● 使う部品

- Arduino
- 抵抗
- LED

#### ● ポイント

- 5V → LED → 抵抗 → GND の順に
- LED には極性があるので向きに注意




図 1: 回路図

### プログラムを書く

LED を接続したピンの電圧を制御するプログラムを書きましょう。

```
import processing.serial.*;
import cc.arduino.*;

Arduino arduino;
int ledPin = 13; // LED を接続したピンの番号

void setup() {
  arduino = new Arduino(this, Arduino.list()[0], 57600);

  // Arduino のピンモードを設定
  // ここでは 13 番ピンを Output 用に設定
  arduino.pinMode(ledPin, Arduino.OUTPUT);
}

void draw() {
  // Arduino の 13 番ピンを HIGH (5V) に
  arduino.digitalWrite(ledPin, Arduino.HIGH);
  delay(500); // 500 ミリ秒間待つ

  // Arduino の 13 番ピンを LOW (0V) に
  arduino.digitalWrite(ledPin, Arduino.LOW);
  delay(500);
}
```

これで LED が点滅するはずです。

## スイッチの ON/OFF を読み取る

Arduino の Digital Input を用いてスイッチの ON/OFF を取得する。

### 回路を組む

デジタル回路の場合、入力端子がどこにも接続されていないような状態 (オープン) が起こると、電圧が High または Low に定まらず誤動作の原因になります。そのため、回路を安定させるためにプルアップ抵抗/プルダウン抵抗と呼ばれる抵抗を用います。




図 2: プルダウン抵抗 (左) とプルアップ抵抗 (右)

### プログラムを書く

```
import processing.serial.*;
import cc.arduino.*;

Arduino arduino;
int switchPin = 8; // スイッチを接続したピンの番号

void setup() {
  size(400, 300);
  arduino = new Arduino(this, Arduino.list()[0], 57600);
  arduino.pinMode(switchPin, Arduino.INPUT); // ピンモードを Input に
}

void draw() {
  // 8 番ピンの電圧を取得し、それが HIGH ならば
  if (arduino.digitalRead(switchPin) == Arduino.HIGH) {
    background(255, 0, 0); // 背景を赤に
  } else {
    background(0, 0, 0); // そうでなければ (LOW ならば) 背景を黒に
  }
}
```

## スイッチを押すと LED が点灯するようにする

スイッチの入力を Processing で取得し、それに基づいて LED を制御しましょう。上 2 つの合わせ技です。

これで入力と出力の両方が実現できるようになります。次回以降の実習でも入力や出力のための部品が変わるだけで基本的な考え方は同じです。

## TRY

今回やったことを思い出しながら、回路とプログラムを作成してみましょう。